Project II

High-Resolution Beamforming on farfield monochromatic signals

Simon Andreas Bjørn

April 13, 2023

Estimating the spatial correlation matrix

We want to estimate the spatial correlation matrix from the data generated by `generate_data.jl`

- Find the spatial correlation matrix by definition

```
R = x*x' / N
heatmap!(ax, abs.(R))
```



Estimate spatial spectrum

We now want to estimate the classical spatial spectrum

- We implement functions for the phase factor, steering vector and DAS.
- Apply `P_DAS` to every angle in DOA

```
# Define functions
DOA  = -40°:0.25°:50°
ϕ(θ) = -k*d*sin.(θ)
a(θ) = @. exp(1im*ϕ(θ)*(0:M-1))
P_DAS(θ) = (a(θ)'*R*a(θ)) / M

# Apply on DAS
P_BF = @. abs(P_DAS(DOA))^2
# Normalize
P_BF ./= maximum(P_BF);
```



Estimate spatial spectrum with Capon's method

Then we calculate it using minimum variance

- Implementing the function is streight forward, following the definition of Capon's method

```
P_CAP(θ) = 1 / (a(θ)'*inv(R)*a(θ))

P_BF2 = @. abs(P_CAP(DOA))^2
P_BF2 ./= maximum(P_BF2);
```



Eigenvalue distribution

We want to find the distribution of eigenvalues and plot them in descending order.

- We decompose **R** using `eigvals` and `eigvecs`.
- They are sorted in ascending order, so we reverse them inplace.

```
dd, V = eigvals(R), eigvecs(R);
reverse!(dd); reverse!(V, dims=2);
```



Spatial spectrum using MUSIC

We assume to know the number of sources is know (2) and we find the spatial spectrum using the MUSIC method.

- Take all eigenvectors corresponding to noise space
- Make MUSIC

```
U = V[:, 3:end]
Π = U*U'
P_M(θ) = 1/(a(θ)'*Π*a(θ))

P_BF3 = @. abs(P_M(DOA))^2
P_BF3 ./= maximum(P_BF3);
```





Spatial spectrum using Eigenvector method

Now we use the eigenvalues as weights in the MUSIC method. This gives the Eigenvector method.

- We construct a diagonal matrix of the Eigenvalues Λ which serve as weights for the Eigenvectors.

```
U = V[:, 3:end]
Λ⁻¹ = inv(diagm(dd[3:end]))
P_EV(θ) = 1/(a(θ)'*U*Λ⁻¹*U'*a(θ))

P_BF4 = @. abs(P_EV(DOA))^2
P_BF4 ./= maximum(P_BF4);
```



Incorrect estimation of number of sources

We now vary the number of sources we assume and plot both the MUSIC and the Eigenvector method.

- We define a function that returns the spatial spectrum given a number of sources
- We see that Eigenvector method given $N = 0$ is identical to Capon's method

```
function assume_sources(N)
  U = V[:, N+1:end]
  Λ⁻¹ = inv(diagm(dd[N+1:end]))
  P0 = θ -> 1/(a(θ)'*U*U'*a(θ))
  P1 = θ -> 1/(a(θ)'*U*Λ⁻¹*U'*a(θ))
  return (
    (@. abs(P0(DOA))^2) |> x -> x ./ maximum(x),
    (@. abs(P1(DOA))^2) |> x -> x ./ maximum(x)
  )
end
```



Coherent sources

Now we modify the `generate_data.jl` to create coherent signals.

- Code is identical to previous tasks, but uses different data.
- There is $30\,\text{deg}$ spearation between the sources compared to only $10\,\text{deg}$ in the previous data.
- We see the classical DAS is now better that all other methods.